
1 Corrections required

1.1 f_TC_22_1_1_Module1
	Function name
	f_TC_22_1_1_Module1

	Reason for change
	In the current TTCN implementation of function f_TC_22_1_1_Module1,  at Step 12b1, in the ATTACH ACCEPT message, the Header Compression Configuration IE is being included.

However, as per the comment in Table 22.1.1.3-7 “the SS does not agree header compression to be applied”. Hence, the TTCN for Step 12b1 is updated to not include the Header Compression Configuration IE. A Prose CR will be raised to address this anomaly in the Table 22.1.1.3-7.

Also in Step 12b1, template cs_EPS_NwkFtSupport_CP_LCS_XtdPCO is being used. This template sets IEs restrictedDCNR and N26Ind to omit. However, as per 24.301 cl 9.9.3.12A, these IEs should be either ‘1’B, or ‘0’B. This needs to be changed.

In the current TTCN implementation of function f_TC_22_1_1_Module1, if the UE does not include APN in the PDN connectivity REQ at Step 4b1, then as a consequence no APN is included in the Activate Default EPS Bearer Context REQ at Step 12b1. This causes error on the UE side.

The APN needs to be included. This change was also pointed out in Anritsu CR R5s180287 Change1, point 3.



	Summary of change
	Updated the TTCN for the ATTACH ACCEPT so that the HeaderCompression Configuration IE is not included, and also updated the template cs_EPS_NwkFtSupport_CP_LCS_XtdPCO. Set the IEs restrictedDCNR and N26Ind to ‘0’B.

Implemented Change1, point 3 in Anritsu CR R5s180287.


	TTCN module
	NBIOT_General_CP.ttcn

	MCC160 Comment
	1. Decision postponed until outcome of prose CR at RAN5
2. Accepted


Before Change:

	<<SKIPPED CODE>>
f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    if (not px_DoAttachWithoutPDN) { // PDN Requested
      if (f_CheckEsmInfoTransferFlag(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST)) {
        //@siclog "Step 11a1" siclog@
        SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                       tsc_SRB1bis,
                                       cs_TimingInfo_Now,
                                       cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cs_508_Esm_Information_Request(f_NBIOT_MobileInfo_GetPTI(0)))));
        //@siclog "Step 11a2" siclog@
        SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                          tsc_SRB1bis,
                                          cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                            cdr_CIOT_Esm_Information_Response(f_NBIOT_MobileInfo_GetPTI(0), cr_ExtdPCOAny))))
          -> value v_ReceivedAsp;
        if (ispresent(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.accessPointName)) {
          v_APN := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.accessPointName;
          v_APN.iei := omit;
          f_NBIOT_MobileInfo_SetAPN (0, v_APN);
        }
        f_NBIOT_MobileInfo_SetExtPCO (0, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.extdProtocolConfigurationOptions); // Table 22.1.1.3-31 specifies mandatory
      }
      v_PDN_Address := f_CIOT_GetPDNAddress(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, f_CheckExtdPCOforIPallocationViaNas(f_NBIOT_MobileInfo_GetExtdPCO(0)));
      v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(f_NBIOT_MobileInfo_GetExtdPCO(0)));
      //@siclog "Step 12b1" siclog@
      SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                     tsc_SRB1bis,
                                     cs_TimingInfo_Now,
                                     cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,
                                                                    cs_CIOT_ATTACH_ACCEPT(cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),
                                                                                          v_Guti,
                                                                                          cs_EPS_NwkFtSupport_CP_LCS_XtdPCO,
                                                                                          cs_AdditionalUpdateResult_CP_SMS,
                                                                                          -,
                                                                                          omit,
                                                                                          omit,
                                                                                          omit),
                                                                    cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,
                                                                                                              f_NBIOT_MobileInfo_GetPTI(0),
                                                                                                              v_APN,
                                                                                                              v_PDN_Address,
                                                                                                              cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),
                                                                                                              -,
                                                                                                              -,
                                                                                                              -,
                                                                                                             v_HeaderCompressConfigTX,

                                                                                                              cs_ControlPlaneInd,
                                                                                                              cs_ServingPLMNRateControl(int2oct (10, 2))))));
      v_ESMMessage_BearerAccept := cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId);
    } else { // No PDN
      //@siclog "Step 12a1" siclog@
      SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                     tsc_SRB1bis,
                                     cs_TimingInfo_Now,
                                     cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,
                                                                    cs_CIOT_ATTACH_ACCEPT(cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),
                                                                                          v_Guti,
                                                                                          cs_EPS_NwkFtSupport_CP_LCS_XtdPCO,
                                                                                          cs_AdditionalUpdateResult_CP_SMS,
                                                                                          -,
                                                                                          omit,
                                                                                          omit,
                                                                                          omit),
                                                                    cs_ESM_DUMMY_MESSAGE(v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg.eSM_DUMMY_MESSAGE.procedureTransactionIdentifier))));
      v_ESMMessage_BearerAccept := cr_ESM_DUMMY_MESSAGE;
    }
<<SKIPPED CODE>>


After Change:

	<<SKIPPED CODE>>

f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    if (not px_DoAttachWithoutPDN) { // PDN Requested
      if (f_CheckEsmInfoTransferFlag(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST)) {
        //@siclog "Step 11a1" siclog@
        SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                       tsc_SRB1bis,
                                       cs_TimingInfo_Now,
                                       cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cs_508_Esm_Information_Request(f_NBIOT_MobileInfo_GetPTI(0)))));
        //@siclog "Step 11a2" siclog@
        SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                          tsc_SRB1bis,
                                          cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                            cdr_CIOT_Esm_Information_Response(f_NBIOT_MobileInfo_GetPTI(0), cr_ExtdPCOAny))))
          -> value v_ReceivedAsp;
        if (ispresent(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.accessPointName)) {
          v_APN := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.accessPointName;
          v_APN.iei := omit;
          f_NBIOT_MobileInfo_SetAPN (0, v_APN);
        }
        f_NBIOT_MobileInfo_SetExtPCO (0, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_INFORMATION_RESPONSE.extdProtocolConfigurationOptions); // Table 22.1.1.3-31 specifies mandatory
      }
      else { //WA#22_1_1 R5s180287
        v_APN := valueof(cs_AccessPointName(px_AccessPointName));
      }
      v_PDN_Address := f_CIOT_GetPDNAddress(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, f_CheckExtdPCOforIPallocationViaNas(f_NBIOT_MobileInfo_GetExtdPCO(0)));
      v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(f_NBIOT_MobileInfo_GetExtdPCO(0)));
      //@siclog "Step 12b1" siclog@
      SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                     tsc_SRB1bis,
                                     cs_TimingInfo_Now,
                                     cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,
                                                                    cs_CIOT_ATTACH_ACCEPT(cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),
                                                                                          v_Guti,
                                                                                          cs_EPS_NwkFtSupport_CP_LCS_XtdPCO,
                                                                                          cs_AdditionalUpdateResult_CP_SMS,
                                                                                          -,
                                                                                          omit,
                                                                                          omit,
                                                                                          omit),
                                                                    cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,
                                                                                                              f_NBIOT_MobileInfo_GetPTI(0),
                                                                                                              v_APN,
                                                                                                              v_PDN_Address,
                                                                                                              cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),
                                                                                                              -,
                                                                                                              -,
                                                                                                              -,
                                                                                                              -, //WA#22_1_1 R&S change v_HeaderCompressConfigTX,
                                                                                                              cs_ControlPlaneInd,
                                                                                                              cs_ServingPLMNRateControl(int2oct (10, 2))))));
      v_ESMMessage_BearerAccept := cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId);
    } else { // No PDN
      //@siclog "Step 12a1" siclog@
      SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                     tsc_SRB1bis,
                                     cs_TimingInfo_Now,
                                     cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,
                                                                    cs_CIOT_ATTACH_ACCEPT(cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),
                                                                                          v_Guti,
                                                                                          cs_EPS_NwkFtSupport_CP_LCS_XtdPCO,
                                                                                          cs_AdditionalUpdateResult_CP_SMS,
                                                                                          -,
                                                                                          omit,
                                                                                          omit,
                                                                                          omit),
                                                                    cs_ESM_DUMMY_MESSAGE(v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg.eSM_DUMMY_MESSAGE.procedureTransactionIdentifier))));
      v_ESMMessage_BearerAccept := cr_ESM_DUMMY_MESSAGE;
    }
<<SKIPPED CODE>>


1.2 f_TC_22_1_1_Module3
	Function name
	f_TC_22_1_1_Module3

	Reason for change
	In the current TTCN implementation of function f_TC_22_1_1_Module3, the variables v_NAS_RcvMsgContainer_SMS and v_NAS_RcvBitsMsgContainer_SMS are being assigned same values twice. This is after Step 16a4B. This needs to be cleaned up.

In function f_TC_22_1_1_Module3, In the TTCN implementation of Step 16a7, the function f_NBIOT_508Check_CP_MOSMSDataTransfer is called, and parameter v_SMS_DELIVER_Sent.tP_UserData is passed as the expected loopback Data. However, this also contains an octet for padding. This will cause a mnismatch when the UE loops back the data because that would not contain padding. Similar issue is observed at Step 16a8 as well. This needs to be addressed


	Summary of change
	Removed the duplicate assignment for v_NAS_RcvMsgContainer_SMS and v_NAS_RcvBitsMsgContainer_SMS.

Created a new variable v_ExpectedLoopbackData and assigned it to f_GetN_OctetsFromPRBS (0, (124-1)). Then used this variable as the expected data in Steps 16a7 and 16a8.

	TTCN module
	NBIOT_General_CP.ttcn

	MCC160 Comment
	1. Accepted

2. Postponed until resolution of ongoing discussion on 36.509


Before Change:

	<<SKIPPED CODE>>

function f_TC_22_1_1_Module3() runs on NBIOT_PTC
  {
    var  octetstring v_SMS_Message;
    var NB_SRB_COMMON_IND v_ReceivedAsp;
    var octetstring v_Rxd_SMS;
    var NAS_MessageContainer v_NAS_RcvMsgContainer_SMS;
    var template (value) NAS_MessageContainer v_NAS_TxMessageContainer_SMS;
    var CP_PDU_Type v_RcvSMS_CP_PDU;
    var template CP_PDU_Type v_ExpSMS_CP_PDU;
    var bitstring v_NAS_RcvBitsMsgContainer_SMS;
    var B3_Type v_TI_Value;
    var TransactionIdentifier_Type v_MO_TI_FromSS;
    var TransactionIdentifier_Type v_MO_TI_FromUE;
    var O1_Type v_RpMsgRefRcvd;
    var SMS_DELIVER_Type v_SMS_DELIVER_Sent;
    var SMS_SUBMIT_Type v_SMS_SUBMIT_Rcvd;
    var octetstring v_PRBS_SMS := f_GetN_OctetsFromPRBS (0, 124); // 124 fits with the requirement from 36.509 @sic R5s180287 sic@
    var integer v_LoopDelay := 60;
    timer t_LoopDelay;
    //@siclog "Step 16a1" siclog@
    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell50);
    //@siclog "Step 16a1A- 16a1B" siclog@
    f_NBIOT_Activate_TestMode (nbiot_Cell50, tsc_UE_TestLoopMode_TypeH);
    //@siclog "Step 16a2 - 16a3" siclog@
    f_NBIOT_CloseUE_TestLoopModeGH (nbiot_Cell50, tsc_UE_TestLoopMode_TypeH, tsc_TestLoopModeGH_UplinkMode_NAS, 2, v_LoopDelay);
    t_LoopDelay.start(f_NBIOT_SetTimerToleranceMin(nasTimer, int2float(v_LoopDelay)));
    //@siclog "Step 16a4" siclog@
    //The SS transmits a CP-ACK encapsulated in a Downlink NAS transport message to the UE.
    v_SMS_DELIVER_Sent := valueof(cs_SMS_DELIVER_IoT (v_PRBS_SMS)); // data TPDU with random data
    v_SMS_Message := f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (cs_MT_TI0_FromSS,
                                                        cs_CP_UserData_DELIVER_IoT(v_SMS_DELIVER_Sent))); // data TPDU with random data
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                  cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          cs_NAS_MsgContainer_SMS (v_SMS_Message)))));
    //@siclog "Step 16a4A" siclog@
    // The UE transmits a CP-ACK encapsulated in an Uplink NAS transport message.
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                    tsc_SRB1bis,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                               cr_NAS_MsgContainer_SMS (?) )))) -> value v_ReceivedAsp;
    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (cr_MT_TI0_FromUE);
    f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test Case 22.1.1 M3", "Step 16a4A"); // @sic R5s180287 sic@
    //@siclog "Step 16a4B" siclog@
    //Check: Does the UE transmit a CP-DATA containing an RP-ACK RPDU encapsulated in an Uplink NAS transport message?
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                   tsc_SRB1bis,
                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;
    // Store procedure transaction id to be used in SS responses.
    // needs to be decoded with Decvalue !!
    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    // Store procedure transaction id to be used in SS responses.
    // needs to be decoded with Decvalue !!
    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;

    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);

    v_ExpSMS_CP_PDU := cr_CP_DATA_PDU (cr_MT_TI0_FromUE, cr_CP_UserData_RP_ACK_ul);
    f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test Case 22.1.1 M3", "Step 16a4B");
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a4B");
    //@siclog "Step 16a4C" siclog@
    //The SS transmits a CP-ACK encapsulated in a Downlink NAS transport message to the UE.
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (cs_MT_TI0_FromSS)));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a5" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell50);
    //@siclog "Step 16a6" siclog@
    t_LoopDelay.timeout;
    //@siclog "Step 16a7" siclog@
    // 1st repetition
    v_Rxd_SMS := f_NBIOT_508Check_CP_MOSMSDataTransfer (nbiot_Cell50, v_SMS_DELIVER_Sent.tP_UserData); // @sic R5s180287 sic@
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_Rxd_SMS);
    if (decvalue(v_NAS_RcvBitsMsgContainer_SMS, v_RcvSMS_CP_PDU) != 0) { // @sic R5s180287 sic@
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed SMS does not match");
    }
    v_SMS_SUBMIT_Rcvd := v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_UserData.tP_PDU.SMS_SUBMIT; // @sic R5s180287 sic@
    if (not (match (v_SMS_DELIVER_Sent.tP_UserData, v_SMS_SUBMIT_Rcvd.tP_UserData))) {
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed TPDU does not match");
    }
    else {
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a7");
    }
    //@siclog "Step 16a7A" siclog@
    // The SS transmits a CP-ACK encapsulated in a Downlink NAS Transport message.
    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;
    v_MO_TI_FromSS.tI_Flag := '1'B;
    v_MO_TI_FromSS.tI_Value := v_TI_Value;
    v_MO_TI_FromUE.tI_Flag := '0'B;
    v_MO_TI_FromUE.tI_Value := v_TI_Value;
    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (v_MO_TI_FromSS)));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a7B" siclog@
    // The SS transmits a CP-DATA containing an RP-ACK RPDU encapsulated in an Downlink NAS transport message
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (v_MO_TI_FromSS,
                                                                                                cs_CP_UserData_RP_ACK_dl(v_RpMsgRefRcvd))));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a7C" siclog@
    // Check: Does the UE transmit a CP-ACK encapsulated in an Uplink NAS Transport message?
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                   tsc_SRB1bis,
                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;
    v_NAS_RcvMsgContainer_SMS :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (v_MO_TI_FromUE);
    v_RcvSMS_CP_PDU := f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test case 22.1.1", "Step 9C");
    //@siclog "Step 16a8" siclog@
    // 2nd repetition
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                           tsc_SRB1bis,
                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                             cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                               cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;  // @sic R5s180287 sic@
    // Store procedure transaction id to be used in SS responses.
    // needs to be decoded with Decvalue !!
    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    if (decvalue(v_NAS_RcvBitsMsgContainer_SMS, v_RcvSMS_CP_PDU) != 0) { // @sic R5s180287 sic@
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed SMS does not match");
    }
    v_SMS_SUBMIT_Rcvd := v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_UserData.tP_PDU.SMS_SUBMIT; // @sic R5s180287 sic@
    if (not (match (v_SMS_DELIVER_Sent.tP_UserData, v_SMS_SUBMIT_Rcvd.tP_UserData))) {
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed TPDU does not match");
    }
    else {
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a8");
    }
    // Note TI value used by the UE and set TI variables accordingly
    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;
    v_MO_TI_FromSS.tI_Flag := '1'B;
    v_MO_TI_FromSS.tI_Value := v_TI_Value;
    v_MO_TI_FromUE.tI_Flag := '0'B;
    v_MO_TI_FromUE.tI_Value := v_TI_Value;
    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;
    //@siclog "Step 16a8A" siclog@
    // The SS transmits a CP-ACK encapsulated in a Downlink NAS Transport message.
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (v_MO_TI_FromSS)));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a8B" siclog@
    // The SS transmits a CP-DATA containing an RP-ACK RPDU encapsulated in an Downlink NAS transport message
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (v_MO_TI_FromSS,
                                                                                                cs_CP_UserData_RP_ACK_dl(v_RpMsgRefRcvd))));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a8C" siclog@
    // Check: Does the UE transmit a CP-ACK encapsulated in an Uplink NAS Transport message?
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                   tsc_SRB1bis,
                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;
    v_NAS_RcvMsgContainer_SMS :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (v_MO_TI_FromUE);
    f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test case 22.1.1", "Step 9C");
    //@siclog "Step 16a9" siclog@
    if (px_DoAttachWithoutPDN) {
      f_NBIOT_508Check_ReleasePDN_RRCConnected (nbiot_Cell50);
    }
    //@siclog "Step 16a10 - 16a11" siclog@
    f_NBIOT_Deactivate_TestMode (nbiot_Cell50, tsc_SRB1bis);
    //@siclog "Step 16a11" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell50);
    } // End of Module 3
<<SKIPPED CODE>>


After Change:

	<<SKIPPED CODE>>

function f_TC_22_1_1_Module3() runs on NBIOT_PTC
  {
    var  octetstring v_SMS_Message;
    var NB_SRB_COMMON_IND v_ReceivedAsp;
    var octetstring v_Rxd_SMS;
    var NAS_MessageContainer v_NAS_RcvMsgContainer_SMS;
    var template (value) NAS_MessageContainer v_NAS_TxMessageContainer_SMS;
    var CP_PDU_Type v_RcvSMS_CP_PDU;
    var template CP_PDU_Type v_ExpSMS_CP_PDU;
    var bitstring v_NAS_RcvBitsMsgContainer_SMS;
    var B3_Type v_TI_Value;
    var TransactionIdentifier_Type v_MO_TI_FromSS;
    var TransactionIdentifier_Type v_MO_TI_FromUE;
    var O1_Type v_RpMsgRefRcvd;
    var SMS_DELIVER_Type v_SMS_DELIVER_Sent;
    var SMS_SUBMIT_Type v_SMS_SUBMIT_Rcvd;
    var octetstring v_PRBS_SMS := f_GetN_OctetsFromPRBS (0, 124); // 124 fits with the requirement from 36.509 @sic R5s180287 sic@
    var octetstring v_ExpectedLoopbackData := f_GetN_OctetsFromPRBS (0, (124-1)); //WA#22_1_1 Less 1 Octets for padding
    var integer v_LoopDelay := 60;
    timer t_LoopDelay;
    //@siclog "Step 16a1" siclog@
    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell50);
    //@siclog "Step 16a1A- 16a1B" siclog@
    f_NBIOT_Activate_TestMode (nbiot_Cell50, tsc_UE_TestLoopMode_TypeH);
    //@siclog "Step 16a2 - 16a3" siclog@
    f_NBIOT_CloseUE_TestLoopModeGH (nbiot_Cell50, tsc_UE_TestLoopMode_TypeH, tsc_TestLoopModeGH_UplinkMode_NAS, 2, v_LoopDelay);
    t_LoopDelay.start(f_NBIOT_SetTimerToleranceMin(nasTimer, int2float(v_LoopDelay)));
    //@siclog "Step 16a4" siclog@
    //The SS transmits a CP-ACK encapsulated in a Downlink NAS transport message to the UE.
    v_SMS_DELIVER_Sent := valueof(cs_SMS_DELIVER_IoT (v_PRBS_SMS)); // data TPDU with random data
    v_SMS_Message := f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (cs_MT_TI0_FromSS,
                                                        cs_CP_UserData_DELIVER_IoT(v_SMS_DELIVER_Sent))); // data TPDU with random data
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                  cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          cs_NAS_MsgContainer_SMS (v_SMS_Message)))));
    //@siclog "Step 16a4A" siclog@
    // The UE transmits a CP-ACK encapsulated in an Uplink NAS transport message.
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                    tsc_SRB1bis,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                               cr_NAS_MsgContainer_SMS (?) )))) -> value v_ReceivedAsp;
    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (cr_MT_TI0_FromUE);
    f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test Case 22.1.1 M3", "Step 16a4A"); // @sic R5s180287 sic@
    //@siclog "Step 16a4B" siclog@
    //Check: Does the UE transmit a CP-DATA containing an RP-ACK RPDU encapsulated in an Uplink NAS transport message?
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                   tsc_SRB1bis,
                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;
    // Store procedure transaction id to be used in SS responses.
    // needs to be decoded with Decvalue !!
    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    // Store procedure transaction id to be used in SS responses.
    // needs to be decoded with Decvalue !!
    //WA#22_1_1 v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    //WA#22_1_1 v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    v_ExpSMS_CP_PDU := cr_CP_DATA_PDU (cr_MT_TI0_FromUE, cr_CP_UserData_RP_ACK_ul);
    f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test Case 22.1.1 M3", "Step 16a4B");
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a4B");
    //@siclog "Step 16a4C" siclog@
    //The SS transmits a CP-ACK encapsulated in a Downlink NAS transport message to the UE.
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (cs_MT_TI0_FromSS)));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a5" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell50);
    //@siclog "Step 16a6" siclog@
    t_LoopDelay.timeout;
    //@siclog "Step 16a7" siclog@
    // 1st repetition
    v_Rxd_SMS := f_NBIOT_508Check_CP_MOSMSDataTransfer (nbiot_Cell50, v_ExpectedLoopbackData /*WA#22_1_1 v_SMS_DELIVER_Sent.tP_UserData*/); // @sic R5s180287 sic@
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_Rxd_SMS);
    if (decvalue(v_NAS_RcvBitsMsgContainer_SMS, v_RcvSMS_CP_PDU) != 0) { // @sic R5s180287 sic@
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed SMS does not match");
    }
    v_SMS_SUBMIT_Rcvd := v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_UserData.tP_PDU.SMS_SUBMIT; // @sic R5s180287 sic@
    if (not (match (v_ExpectedLoopbackData /*WA#22_1_1 v_SMS_DELIVER_Sent.tP_UserData*/, v_SMS_SUBMIT_Rcvd.tP_UserData))) {
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed TPDU does not match");
    }
    else {
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a7");
    }
    //@siclog "Step 16a7A" siclog@
    // The SS transmits a CP-ACK encapsulated in a Downlink NAS Transport message.
    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;
    v_MO_TI_FromSS.tI_Flag := '1'B;
    v_MO_TI_FromSS.tI_Value := v_TI_Value;
    v_MO_TI_FromUE.tI_Flag := '0'B;
    v_MO_TI_FromUE.tI_Value := v_TI_Value;
    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (v_MO_TI_FromSS)));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a7B" siclog@
    // The SS transmits a CP-DATA containing an RP-ACK RPDU encapsulated in an Downlink NAS transport message
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (v_MO_TI_FromSS,
                                                                                                cs_CP_UserData_RP_ACK_dl(v_RpMsgRefRcvd))));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a7C" siclog@
    // Check: Does the UE transmit a CP-ACK encapsulated in an Uplink NAS Transport message?
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                   tsc_SRB1bis,
                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;
    v_NAS_RcvMsgContainer_SMS :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (v_MO_TI_FromUE);
    v_RcvSMS_CP_PDU := f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test case 22.1.1", "Step 9C");
    //@siclog "Step 16a8" siclog@
    // 2nd repetition
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                           tsc_SRB1bis,
                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                             cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                               cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;  // @sic R5s180287 sic@
    // Store procedure transaction id to be used in SS responses.
    // needs to be decoded with Decvalue !!
    v_NAS_RcvMsgContainer_SMS := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    if (decvalue(v_NAS_RcvBitsMsgContainer_SMS, v_RcvSMS_CP_PDU) != 0) { // @sic R5s180287 sic@
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed SMS does not match");
    }
    v_SMS_SUBMIT_Rcvd := v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_UserData.tP_PDU.SMS_SUBMIT; // @sic R5s180287 sic@
    if (not (match (v_ExpectedLoopbackData /*WA#22_1_1v_SMS_DELIVER_Sent.tP_UserData*/, v_SMS_SUBMIT_Rcvd.tP_UserData))) {
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Loop backed TPDU does not match");
    }
    else {
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 16a8");
    }
    // Note TI value used by the UE and set TI variables accordingly
    v_TI_Value := v_RcvSMS_CP_PDU.CP_DATA.transactionIdentifier.tI_Value;
    v_MO_TI_FromSS.tI_Flag := '1'B;
    v_MO_TI_FromSS.tI_Value := v_TI_Value;
    v_MO_TI_FromUE.tI_Flag := '0'B;
    v_MO_TI_FromUE.tI_Value := v_TI_Value;
    v_RpMsgRefRcvd :=  v_RcvSMS_CP_PDU.CP_DATA.cP_UserData.rP_PDU.RP_DATA_ul.rP_MessageReference;
    //@siclog "Step 16a8A" siclog@
    // The SS transmits a CP-ACK encapsulated in a Downlink NAS Transport message.
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_ACK_PDU (v_MO_TI_FromSS)));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a8B" siclog@
    // The SS transmits a CP-DATA containing an RP-ACK RPDU encapsulated in an Downlink NAS transport message
    v_NAS_TxMessageContainer_SMS := cs_NAS_MsgContainer_SMS (f_SMS_PDU_Encvalue(cs_CP_DATA_PDU (v_MO_TI_FromSS,
                                                                                                cs_CP_UserData_RP_ACK_dl(v_RpMsgRefRcvd))));
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell50,
                                   tsc_SRB1bis,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request (tsc_SHT_IntegrityProtected_Ciphered,
                                                 cs_SMS_DL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                          v_NAS_TxMessageContainer_SMS))));
    //@siclog "Step 16a8C" siclog@
    // Check: Does the UE transmit a CP-ACK encapsulated in an Uplink NAS Transport message?
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell50,
                                   tsc_SRB1bis,
                                  cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cr_SMS_UL_NAS_Transport (tsc_SHT_NoSecurityProtection,
                                                                             cr_NAS_MsgContainer_SMS (?))))) -> value v_ReceivedAsp;
    v_NAS_RcvMsgContainer_SMS :=  v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_NAS_TRANSPORT.nasMessage;
    v_NAS_RcvBitsMsgContainer_SMS := oct2bit(v_NAS_RcvMsgContainer_SMS.smsmessage);
    v_ExpSMS_CP_PDU := cr_CP_ACK_PDU (v_MO_TI_FromUE);
    f_SMS_CP_PDU_DecodeAndMatch(v_NAS_RcvBitsMsgContainer_SMS, v_ExpSMS_CP_PDU, "Test case 22.1.1", "Step 9C");
    //@siclog "Step 16a9" siclog@
    if (px_DoAttachWithoutPDN) {
      f_NBIOT_508Check_ReleasePDN_RRCConnected (nbiot_Cell50);
    }
    //@siclog "Step 16a10 - 16a11" siclog@
    f_NBIOT_Deactivate_TestMode (nbiot_Cell50, tsc_SRB1bis);
    //@siclog "Step 16a11" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell50);
    } // End of Module 3
<<SKIPPED CODE>>


1.3 cs_EPS_NwkFtSupport_CP_LCS_XtdPCO
	Function name
	cs_EPS_NwkFtSupport_CP_LCS_XtdPCO

	Reason for change
	This template sets IEs restrictedDCNR and N26Ind to omit. However, as per 24.301 cl 9.9.3.12A, these IEs should be either ‘1’B, or ‘0’B. This needs to be changed

	Summary of change
	Updated the template cs_EPS_NwkFtSupport_CP_LCS_XtdPCO. Set the IEs restrictedDCNR and N26Ind to ‘0’B 

	TTCN module
	NBIOT_General_CP.ttcn

	MCC160 Comment
	Accepted


Before Change:

	template (value) EPS_NetworkFeatureSupport cs_EPS_NwkFtSupport_CP_LCS_XtdPCO :=
    cs_EPS_NwkFtSup_Common (-, -, '1'B, -, -, '02'O, '1'B, '1'B, tsc_Spare1, '0'B, '1'B, '1'B, '0'B, '0'B);


After Change:

	template (value) EPS_NetworkFeatureSupport cs_EPS_NwkFtSupport_CP_LCS_XtdPCO :=
    cs_EPS_NwkFtSup_Common (-, -, '1'B, -, -, '02'O, '1'B, '1'B, tsc_Spare1, '0'B, '1'B, '1'B, '0'B, '0'B, '0'B, '0'B /*WA#22_1_1*/);


